Math 3130 - Introduction to Differential Equations

A Bifurcation Example from Class

```
> restart;
  with(plots):
  with(DEtools):
```

First, we'll define our one parameter family of equations: $y' = y^3 - 2\,y^2 + y + \mu$.

```
> DE1 := diff(y(t),t) = y(t)^3-2*y(t)^2+y(t)+mu;
```

$$DE1 := \frac{d}{dt}\,y(t) = y(t)^3 - 2\,y(t)^2 + y(t) + \mu \tag{1}$$

It turns out that the bifurcation values are μ = 0 and μ = -4/27. If we use the symbolic solver to solve our equation with μ values around and at the bifurcation values, we can see the formula for the solution look a little different. But overall, this demonstrates that the symbolic solver doesn't really help us see any kind of long term or large scale structure of solutions...

```
> dsolve(DE1,y(t));
  dsolve(subs(mu=-1,DE1),y(t));
  dsolve(subs(mu=-4/27,DE1),y(t));
  dsolve(subs(mu=-1/10,DE1),y(t));
  dsolve(subs(mu=0,DE1),y(t));
  dsolve(subs(mu=1,DE1),y(t));
```

$$t - \left(\int^{y(t)} \frac{1}{\_a^3 - 2\,\_a^2 + \_a + \mu}\,d\_a\right) + \_C1 = 0$$

$$t - \left(\int^{y(t)} \frac{1}{\_a^3 - 2\,\_a^2 + \_a - 1}\,d\_a\right) + \_C1 = 0$$

$$y(t) = \frac{e^{RootOf\left(\ln\left(e^{\_Z} + 3\right) e^{\_Z} + \_C1\, e^{\_Z} - \_Z e^{\_Z} + t\, e^{\_Z} + 3 \ln\left(e^{\_Z} + 3\right) + 3\,\_C1 - 3\,\_Z + 3\,t - 3\right)}}{3} + \frac{4}{3}$$

$$t - \left(\int^{y(t)} \frac{1}{\_a^3 - 2\,\_a^2 + \_a - \frac{1}{10}}\,d\_a\right) + \_C1 = 0$$

$$y(t) = e^{RootOf\left(-\ln\left(e^{\_Z} + 1\right) e^{\_Z} + \_C1\, e^{\_Z} + \_Z e^{\_Z} + t\, e^{\_Z} + 1\right)} + 1$$

$$t - \left(\int^{y(t)} \frac{1}{\_a^3 - 2\,\_a^2 + \_a + 1}\,d\_a\right) + \_C1 = 0 \tag{2}$$

Below we graph the right hand side of our equation set equal to zero. This plots the equilibrium values for y versus the parameter μ.

In class we decorated this graph adding in vertical lines and arrows to note what the phase lines looked like for various μ values.

```
> implicitplot(0=y^3-2*y^2+y+mu,mu=-0.35..0.2,y=-3..3,numpoints=5000);
```



The above graph indicates a change in behavoir when μ is approximately -0.15 and again when $\mu = 0$. Let's find the exact bifurcation values.

First, we use our linearization results which says behavoir changing comes when the derivative of the right hand side of our equation hits zero. So we look for all y's when this happens.

```
> solve(diff(y^3-2*y^2+y+mu,y)=0);
```

$$1, \frac{1}{3} \tag{3}$$

Next, we need to find which μ values have equilibria of y=1 and y=1/3. Thus we substitute y=1/3 and y=1 into the right hand side of our equation and look for which μ make that zero. This gives us our bifurcation values.

```
> solve(subs(y=1/3,y^3-2*y^2+y+mu)=0);
  solve(subs(y=1,y^3-2*y^2+y+mu)=0);
```

$$-\frac{4}{27}$$
$$0 \tag{4}$$

Here's a different way to produce a bifurcation diagram.

First, define the right hand side of the equation.

```
> f := (mu,y) -> y^3-2*y^2+y+mu:
  'f(mu,y)' = f(mu,y);
```

$$f(\mu, y) = y^3 - 2y^2 + \mu + y \quad (5)$$

Now we use "contourplot" to find where the right hand side of our equation is above and below zero.

We draw a contour at level 0 (thus plotting f(mu,y)=0 in black). Then every f value below zero is shaded in our first color (red) and every value above zero is shaded in our second color (cyan). If we drew vertical (phase) lines, the part of the lines shaded red should have down arrows and the part of the lines in the cyan region should have up arrows.

```
> # Red up to Cyan. Only plot f(mu,y)=0 contour thus red = decrease,
  cyan = increase
  contourplot(f(mu,y), mu=-0.35..0.2, y=-0.3..1.6, filledregions=true,
  coloring = ["Red", "Cyan"], contours=[0]);
```